

Project 1: Optimizing the Performance of a Pipelined Processor

518030910211 Ziqi Zhao bugenzhao@sjtu.edu.cn
518030910188 Yimin Zhao doctormin@sjtu.edu.cn

May 3, 2020

1 Introduction

Part A

In part A, we write three simple assembly programs to implement three functions in `example.c`. Based on ensuring correctness, we especially focus on the functional equivalence with the example C functions. By selecting and placing labels in the assembly code appropriately, the code is also very readable.

Part B

In part B, we modify the `HCL` file of the Y86's sequential design to add a new instruction — `iaddl`. The following is the roadmap to finish this part:

- Clarify the computation process of `iadd` and write it down at the beginning in `seq-full.hcl`.
- Add dependency relations of `iaddl` to all boolean signals.
- Design the datapath for `iaddl`, i.e., generate control signals for `src` and `dst`

Part C

We achieve full scores in the benchmark testing **in just 2 hours**, but we **spent 2 more days** researching all the potential methods to optimize the performance even further. The following is our roadmap:

- Change the order of the instruction sequence to avoid data hazard and structure hazards as much as possible, which leaves $CPE = 12.96$.
- Beyond the changes on instruction order, we apply loop unrolling to reduce the number of conditional check and registers updating, which leaves $CPE = 9.83$.
- Use a binary search tree to find the precise remaining number of elements after several rounds of unrolling to achieve complete unrolling, which leaves $CPE = 8.95$.
- Modify the pipeline design in `HCL` file to achieve 100% accuracy in branch prediction for certain code pattern, which **brings CPE down to 7.79**.

Contribution

- **Ziqi Zhao** : Part A (coding) & Part B (coding) & Part C (coding & designing) & project report (reviewing)
- **Yimin Zhao** : Part A (reviewing & coding) & Part B (reviewing) & Part C (designing) & project report (writing)

Special Notes for Testing the Part C

- In the Part C, we first achieved full marks by simply adding the `iaddl` instruction to `HCL` file, and in an almost software-only way. The hardware implementation of this approach is in `pipe-full.hcl`.
- After that, we made a further exploration and made a modification to the branch prediction part, and finally achieved a CPE of 7.79. [The hardware implementation of this approach is in `pipe-zzcc.hcl`.](#)

2 Experiments

2.1 Part A

2.1.1 Analysis

In this part, we are asked to implement and simulate three Y86 programs. From a macro point of view, this part is relatively easy. But there are plenty of optimizations worth exploring in terms of code readability and elegance.

Difficult Point

- Always pull the correct element from the stack.
- Be careful to protect the callee-saved register: **EBX, EDI, ESI**.
- Implement function recursion smartly.

Core Technique

- Mimicking C functions, division of functional areas with enough and clear label.
- Get the fastest completion speed by coding line by line referring to C language functions.
- Always draw a picture of the stack to ensure the correctness of fetching a variable.

2.1.2 Code

sum.js

```
1 # 518030910211 ZiqiZhao
2 # 518030910188 YiminZhao
3
4     .pos    0
5 main:
6     irmovl  stack, %esp    # initialize stack
7     rrmovl  %esp, %ebp    # initialize frame
8     irmovl  ele1, %eax
9     pushl   %eax           # argument
10    call    sum_list
11    halt
12
13 # Sample linked list
14 .align 4
15 ele1:
16     .long   0x00a
17     .long   ele2
18 ele2:
19     .long   0x0b0
20     .long   ele3
```

```

21 ele3:
22     .long    0xc00
23     .long    0
24
25 # sum_list func
26 sum_list:
27     pushl    %ebp                # enter
28     rrmovl   %esp, %ebp
29     xorl     %eax, %eax          # clear %eax
30     mrmovl   8(%ebp), %edx       # %edx = ls
31     jmp      test
32 loop:
33     mrmovl   (%edx), %ecx        # tmp = ls->val
34     addl     %ecx, %eax          # val += tmp
35     mrmovl   4(%edx), %edx       # ls = ls->next
36 test:
37     andl     %edx, %edx          # ls == 0?
38     jne      loop               # no -> loop
39 return:
40     rrmovl   %ebp, %esp          # leave
41     popl     %ebp
42     ret
43
44
45 # Stack
46     .pos     0x400
47 stack:

```

rsum.js

```

1  # 518030910211 ZiqiZhao
2  # 518030910188 YiminZhao
3
4  # Set up stack
5      .pos     0
6      irmovl   stack, %esp
7      rrmovl   %esp, %ebp
8      irmovl   ele1, %eax
9      pushl    %eax
10     call     rsum_list
11     halt
12
13 # Sample linked list
14 .align 4
15 ele1:
16     .long     0x00a
17     .long     ele2
18 ele2:

```

```

19     .long    0x0b0
20     .long    ele3
21 ele3:
22     .long    0xc00
23     .long    0
24
25 # rsum_list func
26 rsum_list:
27     pushl    %ebp                # enter
28     rrmovl   %esp, %ebp
29     pushl    %ebx                # save %ebx
30     xorl     %eax, %eax          # clear eax
31     mrmovl   8(%ebp), %edx        # get ls
32     andl     %edx, %edx          # ls == NULL?
33     je       return              # yes -> return
34 do:
35     mrmovl   (%edx), %ebx         # mov ls->val to %ebx
36     mrmovl   4(%edx), %eax
37     pushl    %eax                # push ls->next
38     call     rsum_list
39     addl     %ebx, %eax           # ret = val + ret
40     popl     %edx                # eat para
41 return:
42     popl     %ebx                # restore %ebx
43     rrmovl   %ebp, %esp          # leave
44     popl     %ebp
45     ret
46
47
48 # Stack
49     .pos     0x400
50 stack:

```

copy.js

```

1 # 518030910211 ZiqiZhao
2 # 518030910188 YiminZhao
3
4 # Set up stack
5     .pos     0
6     irmovl   stack, %esp
7     rrmovl   %esp, %ebp
8     irmovl   $3, %eax            # len
9     pushl    %eax
10    irmovl   dest, %eax           # dest
11    pushl    %eax
12    irmovl   src, %eax            # src
13    pushl    %eax

```

```

14     call    copy_block
15     halt
16
17     .align 4
18     # Source block
19     src:
20         .long 0x00a
21         .long 0x0b0
22         .long 0xc00
23         .long 0x888             # Should not copy this
24
25     # Destination block
26     dest:
27         .long 0x111
28         .long 0x222
29         .long 0x333
30         .long 0x999             # Should not write this
31
32     copy_block:
33         pushl    %ebp           # enter
34         rrmovl   %esp, %ebp
35         pushl    %ebx           # save %ebx -> len
36         pushl    %edi           # save %edi -> immediate
37         pushl    %esi           # save %esi -> val
38         xorl     %eax, %eax     # %eax = result = 0
39         mrmovl   16(%ebp), %ebx # %ebx = len
40         mrmovl   12(%ebp), %edx # %edx = dest
41         mrmovl   8(%ebp), %ecx  # %ecx = src
42         jmp      test
43     loop:
44         irmovl   $4, %edi       # %edi = 4
45         mrmovl   (%ecx), %esi   # val = *src
46         addl     %edi, %ecx      # src += 1
47         rmmovl   %esi, (%edx)   # *dest = val
48         addl     %edi, %edx      # val += 1
49         xorl     %esi, %eax     # result ^= val
50         irmovl   $-1, %edi      # %edx = -1
51         addl     %edi, %ebx      # len -= 1
52     test:
53         andl     %ebx, %ebx     # len > 0?
54         jg       loop          # yes -> loop
55     return:
56         popl     %esi           # restore registers
57         popl     %edi
58         popl     %ebx
59         rrmovl   %ebp, %esp     # leave
60         popl     %ebp
61         ret

```

```

62 # Stack
63     .pos    0x400
64 stack:

```

2.1.3 Evaluation

- **sum.y**s

```

../yas sum.y
../yis sum.yo
Stopped in 31 steps at PC = 0x15.  Status 'HLT', CC Z=1 S=0 O=0
Changes to registers:
%eax:  0x00000000      0x00000cba
%ecx:  0x00000000      0x00000c00
%esp:  0x00000000      0x000003fc
%ebp:  0x00000000      0x00000400

Changes to memory:
0x03f4: 0x00000000      0x00000400
0x03f8: 0x00000000      0x00000015
0x03fc: 0x00000000      0x00000018

```

Figure 1: Part A: sum.y

- The `%eax` register has the correct value which is the return value of the function — `0xcba`.
- The memory is not corrupted since all the modifications locate at the stack whose starting addresss is set to be `0x400`.

- **rsum.y**s

```

../yas rsum.y
../yis rsum.yo
Stopped in 68 steps at PC = 0x15.  Status 'HLT', CC Z=0 S=0 O=0
Changes to registers:
%eax:  0x00000000      0x00000cba
%edx:  0x00000000      0x00000020
%esp:  0x00000000      0x000003fc
%ebp:  0x00000000      0x00000400

Changes to memory:
0x03c0: 0x00000000      0x00000c00
0x03c4: 0x00000000      0x000003d4
0x03c8: 0x00000000      0x00000058
0x03d0: 0x00000000      0x000000b0
0x03d4: 0x00000000      0x000003e4
0x03d8: 0x00000000      0x00000058
0x03dc: 0x00000000      0x00000028
0x03e0: 0x00000000      0x0000000a
0x03e4: 0x00000000      0x000003f4
0x03e8: 0x00000000      0x00000058
0x03ec: 0x00000000      0x00000020
0x03f4: 0x00000000      0x00000400
0x03f8: 0x00000000      0x00000015
0x03fc: 0x00000000      0x00000018

```

Figure 2: Part A: rsum.y

- The `%eax` register has the correct value which is the return value of the function — `0xcba`.
- The memory is not corrupted since all the modifications locate at the stack whose starting addresss is set to be `0x400`.

- **copy.y**

```

../yas copy.y
../yis copy.yo
Stopped in 58 steps at PC = 0x25.  Status 'HLT', CC Z=1 S=0 O=0
Changes to registers:
%eax: 0x00000000      0x00000cba
%ecx: 0x00000000      0x00000034
%edx: 0x00000000      0x00000044
%esp: 0x00000000      0x000003f4
%ebp: 0x00000000      0x00000400

Changes to memory:
0x0038: 0x00000111      0x0000000a
0x003c: 0x00000222      0x000000b0
0x0040: 0x00000333      0x00000c00
0x03ec: 0x00000000      0x00000400
0x03f0: 0x00000000      0x00000025
0x03f4: 0x00000000      0x00000028
0x03f8: 0x00000000      0x00000038
0x03fc: 0x00000000      0x00000003

```

Figure 3: Part A: copy.y

- The `%eax` register has the correct value which is the return value of the function — `0xcba`.
- Values are written into the memory correctly as shown in the first three rows in the "Changes to memory" part in Figure 3.
- The memory is not corrupted since all the modifications other than `dest` locate at the stack whose starting addresss is set to be `0x400`.

2.2 Part B

2.2.1 Analysis

In part B, we are asked to extend the SEQ processor to support instruction `iaddl` by modifying `SEQ-full.hcl`. The task is not so difficult once we understand the processing logic and the syntax of `HCL`, and all we need to do is change the followings in `seq-full.hcl`:

- Add `IIADDL` in the choices region of `(bool) instr_valid` since `iaddl` is a valid instruction.
- Add `IIADDL` in the choices region of `(bool) need_regid` since `iaddl` operation involves one register.
- Add `IIADDL` in the choices region of `(bool) need_valC` since `iaddl` operation involves one constant (represented by `valC` in the circuit of Y86 SEQ).
- Add `IIADDL` in the choices region of `(bool) set_cc` since `iaddl` operation involves ALU operation which will set flags.
- When icode is `IIADDL`, `alufun` will be `ALUADD` since the operation is "adding" the constant to `rB`.
- When icode is `IIADDL`, `srcB` is from `rB` since the second operand of `iaddl` is a register.
- When icode is `IIADDL`, `dstE` (where the result from ALU is passed towards) is `rB` since "`iaddl constant, rB`" means `rB += constant` (`rB` is updated).
- When icode is `IIADDL`, `aluA` (the first op) is `valC` (the constant in the instruction) since "`iaddl constant, rB`" means the first op is the constant (`valC`).
- When icode is `IIADDL`, `aluB` (the second op) is `valB` (the value of the second register that is read) for the same reason above.

2.2.2 Code

Modifications in `SEQ-full.hcl`

```
1 -----
2 bool instr_valid = icode in
3   { INOP, IHALT, IRRMOVL, IIRMOVL, IRMMOVL, IMRMOVL,
4     IOPL, IJXX, ICALL, IRET, IPUSHL, IPOPL, IIADDL };
5 -----
6 # Does fetched instruction require a regid byte?
7 bool need_regids =
8   icode in { IRRMOVL, IOPL, IPUSHL, IPOPL,
9             IIRMOVL, IRMMOVL, IMRMOVL, IIADDL };
10 -----
11 # Does fetched instruction require a constant word?
12 bool need_valC =
13   icode in { IIRMOVL, IRMMOVL, IMRMOVL, IJXX, ICALL, IIADDL };
14 -----
```

```

15 ## What register should be used as the B source?
16 int srcB = [
17     icode in { IOPL, IRMMOVL, IMRMOVL, IIADDL } : rB;
18     icode in { IPUSHL, IPOPL, ICALL, IRET } : RESP;
19     1 : RNONE; # Don't need register
20 ];
21 -----
22 ## What register should be used as the E destination?
23 int dstE = [
24     icode in { IRRMOVL } && Cnd : rB;
25     icode in { IIRMOVL, IOPL, IIADDL } : rB;
26     icode in { IPUSHL, IPOPL, ICALL, IRET } : RESP;
27     1 : RNONE; # Don't write any register
28 ];
29 -----
30 ## Select input A to ALU
31 int aluA = [
32     icode in { IRRMOVL, IOPL } : valA;
33     icode in { IIRMOVL, IRMMOVL, IMRMOVL, IIADDL } : valC;
34     icode in { ICALL, IPUSHL } : -4;
35     icode in { IRET, IPOPL } : 4;
36     # Other instructions don't need ALU
37 ];
38 -----
39 ## Select input B to ALU
40 int aluB = [
41     icode in { IRMMOVL, IMRMOVL, IOPL, ICALL,
42               IPUSHL, IRET, IPOPL, IIADDL } : valB;
43     icode in { IRRMOVL, IIRMOVL } : 0;
44     # Other instructions don't need ALU
45 ];
46 -----
47 ## Set the ALU function
48 int alufun = [
49     icode == IOPL : ifun;
50     icode == IIADDL : ALUADD;
51     1 : ALUADD;
52 ];
53 -----
54 ## Should the condition codes be updated?
55 bool set_cc = icode in { IOPL, IIADDL };
56 -----

```

2.2.3 Evaluation

```
→ seq git:(master) ✕ cd ../y86-code && make testssim
../seq/ssim -t asum.yo > asum.seq
../seq/ssim -t asumr.yo > asumr.seq
../seq/ssim -t cjr.yo > cjr.seq
../seq/ssim -t j-cc.yo > j-cc.seq
../seq/ssim -t poptest.yo > poptest.seq
../seq/ssim -t pushquestion.yo > pushquestion.seq
../seq/ssim -t pushtest.yo > pushtest.seq
../seq/ssim -t prog1.yo > prog1.seq
../seq/ssim -t prog2.yo > prog2.seq
../seq/ssim -t prog3.yo > prog3.seq
../seq/ssim -t prog4.yo > prog4.seq
../seq/ssim -t prog5.yo > prog5.seq
../seq/ssim -t prog6.yo > prog6.seq
../seq/ssim -t prog7.yo > prog7.seq
../seq/ssim -t prog8.yo > prog8.seq
../seq/ssim -t ret-hazard.yo > ret-hazard.seq
grep "ISA Check" *.seq
asum.seq:ISA Check Succeeds
asumr.seq:ISA Check Succeeds
cjr.seq:ISA Check Succeeds
j-cc.seq:ISA Check Succeeds
poptest.seq:ISA Check Succeeds
prog1.seq:ISA Check Succeeds
prog2.seq:ISA Check Succeeds
prog3.seq:ISA Check Succeeds
prog4.seq:ISA Check Succeeds
prog5.seq:ISA Check Succeeds
prog6.seq:ISA Check Succeeds
prog7.seq:ISA Check Succeeds
prog8.seq:ISA Check Succeeds
pushquestion.seq:ISA Check Succeeds
pushtest.seq:ISA Check Succeeds
ret-hazard.seq:ISA Check Succeeds
rm asum.seq asumr.seq cjr.seq j-cc.seq poptest.seq pushquestion.seq pushtest.seq prog1.seq prog2.seq prog3.seq prog4.seq prog5.seq prog6.seq prog7.seq prog8.seq ret-hazard.seq
```

Figure 4: Part B: benchmark test

```
→ ptest git:(master) ✕ make SIM=../seq/ssim
./optest.pl -s ../seq/ssim
Simulating with ../seq/ssim
All 49 ISA Checks Succeed
./jtest.pl -s ../seq/ssim
Simulating with ../seq/ssim
All 64 ISA Checks Succeed
./ctest.pl -s ../seq/ssim
Simulating with ../seq/ssim
All 22 ISA Checks Succeed
./htest.pl -s ../seq/ssim
Simulating with ../seq/ssim
All 600 ISA Checks Succeed
→ ptest git:(master) ✕
```

Figure 5: Part B: regression test

```
→ ptest git:(master) ✕ cd ../ptest && make SIM=../seq/ssim TFLAGS=-i
./optest.pl -s ../seq/ssim -i
Simulating with ../seq/ssim
All 58 ISA Checks Succeed
./jtest.pl -s ../seq/ssim -i
Simulating with ../seq/ssim
All 96 ISA Checks Succeed
./ctest.pl -s ../seq/ssim -i
Simulating with ../seq/ssim
All 22 ISA Checks Succeed
./htest.pl -s ../seq/ssim -i
Simulating with ../seq/ssim
All 756 ISA Checks Succeed
→ ptest git:(master) ✕
```

Figure 6: Part B: iaddl test

2.3 Part C

2.3.1 Analysis

In this part, we were asked to speed up the program `ncopy.y` as much as possible by modifying the `ncopy.y` and `HCL`. The following is our roadmap:

Added `iaddl` instruction to `pipe-full.hcl`

Like what we have done in Part B, we added `iaddl` to the pipeline design. It avoided the overhead of using registers to store constants, and also increased the number of registers we could use, which paved the way for our further optimization.

Avoid Load and Use: CPE \rightarrow 12.96

For the pipeline design in CS:APP 2e, "load and use" or `mrmovl` then `rmmovl` will cause penalty, which must be avoided to improve the performance. On the one hand, we rearranged the order of instructions to avoid stalling as much as possible. On the other hand, we use two registers to store the variable `val`, loading them separately and ahead of time.

10-way Loop Unrolling: CPE \rightarrow 9.83

There's much overhead in testing and updating procedure of loops, and one way to minimize it is to perform a technique named "loop unrolling". That is, we do multiple loops and update the relevant data at once, to reduce the number of times we execute the `add` and `jxx` instructions.

Search Tree for Remaining Elements: CPE \rightarrow 8.95

For large inputs, the more ways we unroll the loops, the better the program performs. However, for small inputs, it is important to choose a good method to process the remaining elements. The simplest way is to write another loop for them, but a much better way is to totally unroll the code, that is, jump to different position for different number of remaining ones. Since Y86 does not support relative jump instruction, we designed a search tree to get the correct jump destination for each possibility.

The above optimization took us two hours, so far we have reached full marks.

But can it be even faster? We spent another two days poring over the implementation logic in `HCL` and other files of the Y86 pipeline design. And it finally got us here:

Optimized for Our Branch Prediction Design: CPE \rightarrow 7.79

For this program, a significant performance factor is the branch prediction failure for `count++`. In `pipe-zzcc.hcl`, we made a special optimization for the situation like this:

Instruction	any instruction	non-alu instruction	<code>jxx</code>
Stages	EX	ID	IF

Note that in this case, we can forward the conditions from EX stage to IF and predict the branch with 100% accuracy. Thus, we optimized the program to ensure that there were as many of these patterns as possible and took much advantage of it, which leads to an average CPE of 7.79.

2.3.2 Code

—10-Way Loop Unrolling—

```
1 #####
2 # You can modify this portion
3 # Entry
4     iaddl $-9, %edx        # len -= 9, i.e., initial_len <= 9?
5     irmovl $0, %eax       # count = 0
6     jle Remaining        # if so, goto Remaining
7
8 # Loop unrolling part
9 Loop0:
10    mrmovl (%ebx), %esi    # valA = src[0]
11    mrmovl 4(%ebx), %edi   # valB = src[1]
12    andl %esi, %esi       # valA <= 0?
13    rmmovl %esi, (%ecx)    # dst[0] = valA
14    jle Loop1            # if so, goto next loop
15    iaddl $1, %eax        # count++
16 Loop1:
17    mrmovl 8(%ebx), %esi   # valA = src[2]
18    andl %edi, %edi       # valB <= 0?
19    rmmovl %edi, 4(%ecx)   # dst[1] = valB
20    jle Loop2            # if so, goto next loop
21    iaddl $1, %eax        # count++
22 Loop2:
23    mrmovl 12(%ebx), %edi  # valB = src[3]
24    andl %esi, %esi       # valA <= 0?
25    rmmovl %esi, 8(%ecx)   # dst[2] = valA
26    jle Loop3            # if so, goto next loop
27    iaddl $1, %eax        # count++
28 Loop3:
29    mrmovl 16(%ebx), %esi  # valA = src[4]
30    andl %edi, %edi       # valB <= 0?
31    rmmovl %edi, 12(%ecx)  # dst[3] = valB
32    jle Loop4            # if so, goto next loop
33    iaddl $1, %eax        # count++
34 Loop4:
35    mrmovl 20(%ebx), %edi  # valB = src[5]
36    andl %esi, %esi       # valA <= 0?
37    rmmovl %esi, 16(%ecx)  # dst[4] = valA
38    jle Loop5            # if so, goto next loop
39    iaddl $1, %eax        # count++
40 Loop5:
41    mrmovl 24(%ebx), %esi  # valA = src[6]
42    andl %edi, %edi       # valB <= 0?
43    rmmovl %edi, 20(%ecx)  # dst[5] = valB
44    jle Loop6            # if so, goto next loop
45    iaddl $1, %eax        # count++
```

```

46 Loop6:
47     mrmovl 28(%ebx), %edi    # valB = src[7]
48     andl %esi, %esi         # valA <= 0?
49     rmmovl %esi, 24(%ecx)    # dst[6] = valA
50     jle Loop7               # if so, goto next loop
51     iaddl $1, %eax          # count++
52 Loop7:
53     mrmovl 32(%ebx), %esi    # valA = src[8]
54     andl %edi, %edi         # valB <= 0?
55     rmmovl %edi, 28(%ecx)    # dst[7] = valB
56     jle Loop8               # if so, goto next loop
57     iaddl $1, %eax          # count++
58 Loop8:
59     mrmovl 36(%ebx), %edi    # valB = src[9]
60     andl %esi, %esi         # valA <= 0?
61     rmmovl %esi, 32(%ecx)    # dst[8] = valA
62     jle Loop9               # if so, goto next loop
63     iaddl $1, %eax          # count++
64 Loop9:
65     andl %edi, %edi         # valB <= 0?
66     rmmovl %edi, 36(%ecx)    # dst[9] = valB
67     jle LoopEnd             # if so, goto loop end
68     iaddl $1, %eax
69 LoopEnd:
70     iaddl $40, %ecx          # dst += 10 * 4
71     iaddl $40, %ebx          # src += 10 * 4
72     iaddl $-10, %edx         # len -= 10
73     jg Loop0                 # if so, goto Loop0
74                               # else, goto process remaining elements

```

—Binary Search Tree for Finding the Number of Remaining Loops—

```

1  # The following block is a binary search tree to
2  # find the number of remaining loops
3  # (which must be less than 10) at minimal cost
4  Remaining:
5      iaddl $6, %edx          # [-9,0] -> [-3,6]      (+3)
6  RemTest:
7      irmovl $0, %esi
8      jg RemTestR
9      je Rem3
10 RemTestL:
11     iaddl $2, %edx          # [-3,-1] -> [-1,1]      (+1)
12     je Rem1
13     jg Rem2
14     jmp Done                # -1 + 1 = 0
15 RemTestR:
16     iaddl $-3, %edx         # [1,6] -> [-2,3]      (+6)

```

```

17         jg RemTestRR
18         je Rem6
19 RemTestRL:
20         iaddl $1, %edx          # [-2,-1] -> [-1,0]      (+5)
21         jl Rem4
22         je Rem5
23 RemTestRR:
24         iaddl $-2, %edx        # [1,3] -> [-1,1]        (+8)
25         jl Rem7
26         je Rem8

```

—Unrolling of Remaining Loops—

```

1 Rem9:
2     mrmovl 32(%ebx), %esi      # valA = src[8]
3     rmmovl %esi, 32(%ecx)     # dst[8] = valA
4 Rem8:  # Note that %esi == 0, directly jumping here
5         # implies that RemXb will performs correctly.
6         andl %esi, %esi       # valA <= 0?
7         mrmovl 28(%ebx), %esi  # valA = src[7]
8         jle Rem8b             # if so, goto Rem8b
9         iaddl $1, %eax        # count++
10 Rem8b: rmmovl %esi, 28(%ecx)   # dst[7] = valA
11 Rem7:
12     andl %esi, %esi          # valA <= 0?
13     mrmovl 24(%ebx), %esi     # valA = src[6]
14     jle Rem7b                # if so, goto Rem7b
15     iaddl $1, %eax           # count++
16 Rem7b: rmmovl %esi, 24(%ecx)   # dst[6] = valA
17 Rem6:
18     andl %esi, %esi          # valA <= 0?
19     mrmovl 20(%ebx), %esi     # valA = src[5]
20     jle Rem6b                # if so, goto Rem6b
21     iaddl $1, %eax           # count++
22 Rem6b: rmmovl %esi, 20(%ecx)   # dst[5] = valA
23 Rem5:
24     andl %esi, %esi          # valA <= 0?
25     mrmovl 16(%ebx), %esi     # valA = src[4]
26     jle Rem5b                # if so, goto Rem5b
27     iaddl $1, %eax           # count++
28 Rem5b: rmmovl %esi, 16(%ecx)   # dst[4] = valA
29 Rem4:
30     andl %esi, %esi          # valA <= 0?
31     mrmovl 12(%ebx), %esi     # valA = src[3]
32     jle Rem4b                # if so, goto Rem4b
33     iaddl $1, %eax           # count++
34 Rem4b: rmmovl %esi, 12(%ecx)   # dst[3] = valA
35 Rem3:

```

```

36      andl %esi, %esi      # valA <= 0?
37      mrmovl 8(%ebx), %esi # valA = src[2]
38      jle Rem3b            # if so, goto Rem3b
39      iaddl $1, %eax       # count++
40 Rem3b: rmmovl %esi, 8(%ecx) # dst[2] = valA
41 Rem2:
42      andl %esi, %esi      # valA <= 0?
43      mrmovl 4(%ebx), %esi # valA = src[1]
44      jle Rem2b            # if so, goto Rem2b
45      iaddl $1, %eax       # count++
46 Rem2b: rmmovl %esi, 4(%ecx) # dst[1] = valA
47 Rem1:
48      andl %esi, %esi      # valA <= 0?
49      mrmovl (%ebx), %esi  # valA = src[0]
50      jle Rem1b            # if so, goto Rem1b
51      iaddl $1, %eax       # count++
52 Rem1b:
53      andl %esi, %esi      # valA <= 0?
54      rmmovl %esi, (%ecx)  # dst[0] = valA
55      jle Done             # if so, goto Done
56      iaddl $1, %eax       # count++

```

—Modification to hcl (all in pipe-zzcc.hcl)—

Note: here we have omitted the changes for adding `iaddl`,
which is the same as those in `seq-full.hcl` and `pipe-full.hcl`

```

1  # 1. Added instruction 'iaddl'
2  #   Similar to the changes in '../seq/full.hcl'.
3  #
4  # 2. Optimization on branch prediction
5  #   a. For unconditional JMP, there's no need to insert any bubble.
6  #
7  #   b. A significant performance factor is the branch prediction failure for
8  #      ↪ count++. In our design, we made a special optimization for the
9  #      ↪ situation like this:
10 #
11 #      Instruction:  any    non-alu    jxx
12 #      Stage:       EX     ID         IF
13 #
14 #      Note that in this case, we can forward the conditions from EX stage to
15 #      ↪ IF and predict the branch with 100% accuracy. Thus, we optimized the
16 #      ↪ program to ensure that there were as many of these patterns as possible
17 #      ↪ and took much advantage of it, which leads to an average CPE of 7.79.

```

Specifically, we have modified the following logic:

- Added some additional definitions around line 150;
- Modified SelectPC logic around line 180;
- Modified PredictPC logic around line 230;
- Modified pipeline bubble logic around line 400 and 410.


```

18 -----Added following definition-----
19 quote 'int gen_aluA();'           # Declaration of gen_aluA
20 quote 'int gen_aluB();'           # Declaration of gen_aluB
21
22 # For JXX in ID and ALU in EX, check the cc generated by ALU
23 # Note that the simulator do not generate 'cc_in' correctly
24 boolsig f_cnd_alu
25     'cond_holds(compute_cc(id_ex_curr->ifun,
26                         gen_aluA(), gen_aluB()),
27                 if_id_next->ifun)'
28
29 # For JXX in ID and non-ALU in EX, check the cc register
30 boolsig f_cnd_other 'cond_holds(cc, if_id_next->ifun)'
31
32 -----Modified f_PC-----
33 ## What address should instruction be fetched at
34 int f_pc = [
35     # Unconditional jump: Use predicted value of PC
36     M_icode == IJXX && M_ifun == UNCOND : F_predPC;
37
38     # Mispredicted taken. Fetch at incremented PC (previously valP)
39     M_icode == IJXX && (W_icode in {IOPL, IIADDL}) && !M_Cnd : M_valA;
40
41     # Completion of RET instruction.
42     W_icode == IRET : W_valM;
43
44     # Default: Use predicted value of PC
45     1 : F_predPC;
46 ];
47
48 -----Modified f_predPC-----
49 # Predict next value of PC
50 int f_predPC = [
51     f_icode == ICALL : f_valC;
52     f_icode == IJXX && f_ifun == UNCOND : f_valC;
53
54     # Decode stage is ALU and will set CC->always taken by default
55     f_icode == IJXX && (D_icode in {IOPL, IIADDL}): f_valC;
56
57     # Decode stage is not ALU
58     # Execute stage is ALU -> compute CC
59     f_icode == IJXX && (E_icode in {IOPL, IIADDL}) &&
60     !f_cnd_alu : f_valP;
61
62     # Execute stage is not ALU -> check cc -> ZF SF OF
63     f_icode == IJXX && !(E_icode in {IOPL, IIADDL}) &&
64     !f_cnd_other : f_valP;
65

```

```

66     # Other JXX
67     f_icode == IJXX : f_valC;
68
69     # Otherwise
70     1 : f_valP;
71 l;
72
73 -----Added conditions to D_Bubble & E_Bubble-----
74 bool D_bubble =
75     # Mispredicted branch taken
76     (E_icode == IJXX && E_ifun != UNCOND &&
77      (M_icode in {IOPL, IIADDL}) && !e_Cnd) ||
78     # Stalling at fetch while ret passes through pipeline
79     # but not condition for a load/use hazard
80     !(E_icode in { IMRMOVL, IPOPL }
81      && E_dstM in { d_srcA, d_srcB })
82      && IRET in { D_icode, E_icode, M_icode };
83
84
85 bool E_bubble =
86     # Mispredicted branch taken
87     (E_icode == IJXX && E_ifun != UNCOND &&
88      (M_icode in {IOPL, IIADDL}) && !e_Cnd) ||
89     # Conditions for a load/use hazard
90     E_icode in { IMRMOVL, IPOPL } &&
91     E_dstM in { d_srcA, d_srcB};

```

2.3.3 Evaluation

Note: both designs in `pipe-full.hcl` and `pipe-zzcc.hcl` have passed the tests and got full marks. Considering that the latter is a superset of the former, we have omitted the test screenshots of `pipe-full.hcl` here. The following results are the latter one's.

```

❏ (bz-parallel) sim/pipe git:(master) ► make testpsim
make -C ../ptest SIM=../pipe/psim TFLAGS=-i
make[1]: Entering directory '/home/bugenzhao/ComputerArch-Prj1/sim/ptest'
./optest.pl -s ../pipe/psim -i
Simulating with ../pipe/psim
All 58 ISA Checks Succeed
./jtest.pl -s ../pipe/psim -i
Simulating with ../pipe/psim
All 96 ISA Checks Succeed
./ctest.pl -s ../pipe/psim -i
Simulating with ../pipe/psim
All 22 ISA Checks Succeed
./htest.pl -s ../pipe/psim -i
Simulating with ../pipe/psim
All 756 ISA Checks Succeed
make[1]: Leaving directory '/home/bugenzhao/ComputerArch-Prj1/sim/ptest'

```

Figure 7: Part C: regression test (with `iaddl` included)

```

make -C ../y86-code testpsim
make[1]: Entering directory '/home/bugenzhao/ComputerArch-Prj1/sim/y86-code'
../pipe/psim -t asum.yo > asum.pipe
../pipe/psim -t asumr.yo > asumr.pipe
../pipe/psim -t cjr.yo > cjr.pipe
../pipe/psim -t j-cc.yo > j-cc.pipe
../pipe/psim -t poptest.yo > poptest.pipe
../pipe/psim -t pushquestion.yo > pushquestion.pipe
../pipe/psim -t pushtest.yo > pushtest.pipe
../pipe/psim -t prog1.yo > prog1.pipe
../pipe/psim -t prog2.yo > prog2.pipe
../pipe/psim -t prog3.yo > prog3.pipe
../pipe/psim -t prog4.yo > prog4.pipe
../pipe/psim -t prog5.yo > prog5.pipe
../pipe/psim -t prog6.yo > prog6.pipe
../pipe/psim -t prog7.yo > prog7.pipe
../pipe/psim -t prog8.yo > prog8.pipe
../pipe/psim -t ret-hazard.yo > ret-hazard.pipe
grep "ISA Check" *.pipe
asum.pipe:ISA Check Succeeds
asumr.pipe:ISA Check Succeeds
cjr.pipe:ISA Check Succeeds
j-cc.pipe:ISA Check Succeeds
poptest.pipe:ISA Check Succeeds
prog1.pipe:ISA Check Succeeds
prog2.pipe:ISA Check Succeeds
prog3.pipe:ISA Check Succeeds
prog4.pipe:ISA Check Succeeds
prog5.pipe:ISA Check Succeeds
prog6.pipe:ISA Check Succeeds
prog7.pipe:ISA Check Succeeds
prog8.pipe:ISA Check Succeeds
pushquestion.pipe:ISA Check Succeeds
pushtest.pipe:ISA Check Succeeds
ret-hazard.pipe:ISA Check Succeeds
rm asum.pipe asumr.pipe cjr.pipe j-cc.pipe poptest.pipe pushquestion.pipe push
pe prog8.pipe ret-hazard.pipe
make[1]: Leaving directory '/home/bugenzhao/ComputerArch-Prj1/sim/y86-code'

```

Figure 8: Part C: benchmark test

```

60      OK
61      OK
62      OK
63      OK
64      OK
128     OK
192     OK
256     OK
68/68 pass correctness test

```

Figure 9: Part C: correctness test

58	325	5.60
59	328	5.56
60	336	5.60
61	338	5.54
62	346	5.58
63	345	5.48
64	356	5.56
Average CPE		7.79
Score	60.0/60.0	

Figure 10: Part C: CPE test

3 Conclusion

In this project, we completed the tasks of three parts, which were gradually developed. The first part made us familiar with Y86 assembly syntax, while the second part made us familiar with Y86 SEQ circuit logic, and the third part encouraged us to transform assembly code and Y86 pipeline design. The following is a summary of the completion of the three parts:

- **Part A**

- We write assembly code for three simple functions.
- We take care to protect the stack and registers.
- We focus on the readability and functional equivalence of the code.

- **Part B**

- We modify `SEQ-full.hcl` to add an instruction: `iaddl`.

- **Part C**

- We reorder the instructions to avoid hazards.
- We do 10-way loop unrolling to speed up the `while` loop.
- We create a search tree to find the number of remaining loops at the minimal cost and then completely unroll the operations for remaining elements.
- We modify the `HCL` file of the pipeline to optimize the branch prediction, achieving 100% accuracy for a certain pattern (non-ALU followed by `jxx`).

3.1 Problems

We've only had some tricky problems in Part C. They are two unsuccessful attempts to modify the pipeline logic to lift the accuracy of branch prediction. Based on the second attempt, however, we have made some small changes to achieve the goal, which is explained in detail in **3.2 Achievements**.

3.1.1 Attempt 1

In this attempt, we look at a particular code distribution, which is "`andl op1, op2 → jxx dest`". We hope that the branch prediction of `jxx` under this distribution can reach 100% accuracy.

The logic is shown in Figure 11:




Figure 11: Part C: Attempt 1

Our initial plan is that, when we detect a `jxx` after an `andl`, we will do branch prediction according to the value of the first operand of `andl`. Since in our `ncopy.js`, there are a lot of this code patterns:

```

Loop0:
    mrmovl (%ebx), %esi # valA = src[0]
    mrmovl 4(%ebx), %edi # valB = src[1]
    rmmovl %esi, (%ecx) # dst[0] = valA
    andl %esi, %esi     # valA ≤ 0?
    jle Loop1           # if so, goto next loop
    iaddl $1, %eax      # count++
Loop1:
    mrmovl 8(%ebx), %esi # valA = src[2]
    rmmovl %edi, 4(%ecx) # dst[1] = valB
    andl %edi, %edi     # valB ≤ 0?
    jle Loop2           # if so, goto next loop
    iaddl $1, %eax      # count++
Loop2:
    mrmovl 12(%ebx), %edi # valB = src[3]
    rmmovl %esi, 8(%ecx) # dst[2] = valA
    andl %esi, %esi     # valA ≤ 0?
    jle Loop3           # if so, goto next loop
    iaddl $1, %eax      # count++

```

Figure 12: Part C: Loop of Attempt 1

Let's take a closer look at "`andl op1, op2 → jxx dest`".

When `op1 == op2`, i.e., `rA == rB`, the sign of `op1` is exactly the sign of `op1 & op2`, then we can compare the current conditional jump instruction type with the sign of `op1` to achieve a 100% accuracy.

But if `op1 != op2`, our prediction will not always succeed, which requires a restore from mispredicted branch, that is, we must find out an approach to determine whether we have mispredicted when the `jxx` is in MEM stage. However, this is unsolvable since

the this information of two register sources has been lost in the pipeline when `andl` has entered its WB stage.

3.1.2 Attempt 2

In this attempt, we look at another code pattern, which is "any instruction \rightarrow non-alu instruction \rightarrow `jxx dest`", just like the following:

```

    andl %edi, %edi      # valB ≤ 0?
    rmmovl %edi, 20(%ecx) # dst[5] = valB
    jle Loop6           # if so, goto next loop
    iaddl $1, %eax       # count++
Loop6:
    mrmovl 28(%ebx), %edi # valB = src[7]

```

Figure 13: Part C: Attempt 2

Since the last instruction of `jxx` is a non-alu instruction, we are able to get the correct conditional flag when `jxx` is still in the Fetch stage.

Specifically, if the instruction in Execute stage now is a non-alu one, then the current value of CC register will be the correct one for `jxx`. If the instruction in Execute stage now is an alu one, the the condition flags generated by ALU, i.e. in `cc_in`, will be the correct one.

We have ensured that if a non-alu instruction is followed by a `jxx`, the prediction must be correct. Thus, when we are processing with the logic of restoring from misprediction, we only need to focus on the case where the previous instruction of `jxx` is an alu instruction. According to this, we avoid the problems we encountered in our first attempt, since it is really easy to get the `ifun` field and decide the type of an instruction in Write-back stage.

However, when we finished the implementation in HCL file, **a strange problem arose**. The simulator using this branch prediction approach can pass all the ISA tests provided, but yielded a large number of "bad count" in `ncopy.ys`.

It took us about six hours to troubleshoot the problem and finally pin down the cause of the problem into two signals in the pipeline simulator: **`cc_in` and `e_valE`**. Please refer to the next section on how we finally solved this problem.

3.2 Achievements

3.2.1 A Successful Attempt 3

For the "bad count" problems, we explored for a long time and finally realized that when we predicted the branch, **the simulator may not generate the correct ALU output values**, say, `e_valE` and `cc_in`. After a talk with Chi Zhang, we adopted a compromise method, that is, manually call the C function in the simulator program to generate a correct ALU output, and take this as the basis of prediction. It finally succeeded and led to a CPE of 7.79.

```

1 # For JXX in ID and ALU in EX, check the cc generated by ALU
2 # Note that the simulator do not generate 'cc_in' correctly
3 boolsig f_cnd_alu

```

```

4      'cond_holds(compute_cc(id_ex_curr->ifun,
5                  gen_aluA(), gen_aluB()),
6                  if_id_next->ifun)'

```

In this case, we seem to predict the branch by doing complex operations in the Fetch stage, which also seems unreasonable in real pipelined processor design. However, according to the characteristics of the logic circuit, all signals in the circuit must be able to reach steady state in one clock cycle, so the output of ALU must be correct. **We don't need to add any additional complex hardware in the Fetch stage to perform such a prediction policy. We just need a simple unit instead to compare the ifun of the current jxx instruction with the three flags of zf, sf and of in cc or cc_in**, which is similar to the technique named "delayed slot" and is a fairly practical design in fact.

3.2.2 Performance Improvement

By modifying the pipeline logic, we managed to accelerate the program to a very surprising degree:

$$\text{CPE} = 12.98 \rightarrow \text{CPE} = 9.83 \rightarrow \text{CPE} = 8.95 \rightarrow \text{CPE} = 7.79$$

We even try to unlimit the number of the array size. By modifying `benchmark.pl`, we push the upper bound to be 400 to test the best performance of our implementation (see Figure 14).

390	1953	5.01
391	1955	5.00
392	1963	5.01
393	1962	4.99
394	1973	5.01
395	1980	5.01
396	1980	5.00
397	1983	4.99
398	1991	5.00
399	1994	5.00
400	2002	5.00
Average CPE		5.55
Score	60.0/60.0	

Figure 14: Part C: larger scale test

From analysis, we could safely estimate that the theoretical minimum CPE should be around 4.5 to 5.0. **So it's almost certain that the optimizations we've done under the existing ISA framework are close to extreme.**

3.2.3 Code readability

We put a lot of effort into the readability of the code for parts A.

- For functions that contain loops, we can always break it down into three logical regions: `loop`, `test`, and `return`.
- For various registers, we always annotate its purpose with comments
- For code that's not so obvious, we always leave the comment showing its counterpart in the C function next to it.

We have also left understandable and sufficient comments in the header of the modified files in Part B and Part C. Please refer to `ncopy.js`, `pipe-full.hcl`, `pipe-zzcc.hcl`, and `seq-full.hcl`.

3.3 Feelings

This is a very interesting and valuable project, and we have worked together perfectly. When Ziqi achieved full score for Part C in two hours, we started thinking about whether we could get CPE down to the limitation. We ended up spending two days talking, drawing, and writing code to test. But the most exciting thing was that after two big failures, we finally got there in a very "weird" way. It should be noted, however, that there is an inherent problem with the simulator for the pipeline processor. Otherwise our second attempt should have been successful and would not have required special means to implement it in the third attempt.

This project allowed us to reap the valuable sense of accomplishment brought by not giving up. We are also very grateful to have been able to take this course and to be introduced to such a valuable project.

Finally, we would like to specially thank Miss Shen and the TAs for their instruction and support, so that we can successfully complete this project.